Title

stata.com

ivtobit — Tobit model with continuous endogenous regressors

Syntax

Maximum likelihood estimator

`ivtobit` *depvar* [*varlist*$_1$] (*varlist*$_2$ = *varlist*$_{iv}$) [*if*] [*in*] [*weight*] ,

`ll`[(#)] `ul`[(#)] [*mle_options*]

Two-step estimator

`ivtobit` *depvar* [*varlist*$_1$] (*varlist*$_2$ = *varlist*$_{iv}$) [*if*] [*in*] [*weight*] , `twostep`

`ll`[(#)] `ul`[(#)] [*tse_options*]

mle_options	Description
Model	
* `ll`[(#)]	lower limit for left censoring
* `ul`[(#)]	upper limit for right censoring
`mle`	use conditional maximum-likelihood estimator; the default
`constraints`(*constraints*)	apply specified linear constraints
SE/Robust	
`vce`(*vcetype*)	*vcetype* may be `oim`, `robust`, `cluster` *clustvar*, `opg`, `bootstrap`, or `jackknife`
Reporting	
`level(#)`	set confidence level; default is `level(95)`
`first`	report first-stage regression
`nocnsreport`	do not display constraints
display_options	control column formats, row spacing, line width, display of omitted variables and base and empty cells, and factor-variable labeling
Maximization	
maximize_options	control the maximization process
`coeflegend`	display legend instead of statistics

*You must specify at least one of `ll`[(#)] and `ul`[(#)].

tse_options	Description
Model	
* twostep	use Newey's two-step estimator; the default is mle
* ll[(*#*)]	lower limit for left censoring
* ul[(*#*)]	upper limit for right censoring
SE	
vce(*vcetype*)	*vcetype* may be twostep, bootstrap, or jackknife
Reporting	
level(*#*)	set confidence level; default is level(95)
first	report first-stage regression
display_options	control column formats, row spacing, line width, display of omitted variables and base and empty cells, and factor-variable labeling
coeflegend	display legend instead of statistics

* twostep is required. You must specify at least one of ll[(*#*)] and ul[(*#*)].

varlist$_1$ and *varlist*$_{\rm iv}$ may contain factor variables; see [U] **11.4.3 Factor variables**.

depvar, *varlist*$_1$, *varlist*$_2$, and *varlist*$_{\rm iv}$ may contain time-series operators; see [U] **11.4.4 Time-series varlists**.

bootstrap, by, jackknife, rolling, statsby, and svy are allowed; see [U] **11.1.10 Prefix commands**. fp is allowed with the maximum likelihood estimator.

Weights are not allowed with the bootstrap prefix; see [R] **bootstrap**.

vce(), first, twostep, and weights are not allowed with the svy prefix; see [SVY] **svy**.

fweights, iweights, and pweights are allowed with the maximum likelihood estimator. fweights are allowed with Newey's two-step estimator. See [U] **11.1.6 weight**.

coeflegend does not appear in the dialog box.

See [U] **20 Estimation and postestimation commands** for more capabilities of estimation commands.

Menu

Statistics > Endogenous covariates > Tobit model with endogenous covariates

Description

ivtobit fits tobit models where one or more of the regressors is endogenously determined. By default, ivtobit uses maximum likelihood estimation. Alternatively, Newey's (1987) minimum chi-squared estimator can be invoked with the twostep option. Both estimators assume that the endogenous regressors are continuous and so are not appropriate for use with discrete endogenous regressors. See [R] **ivprobit** for probit estimation with endogenous regressors and [R] **tobit** for tobit estimation when the model contains no endogenous regressors.

Options for ML estimator

Model

`ll(#)` and `ul(#)` indicate the lower and upper limits for censoring, respectively. You may specify one or both. Observations with *depvar* $\leq$ `ll()` are left-censored; observations with *depvar* $\geq$ `ul()` are right-censored; and remaining observations are not censored. You do not have to specify the censoring values at all. It is enough to type `ll`, `ul`, or both. When you do not specify a censoring value, `ivtobit` assumes that the lower limit is the minimum observed in the data (if `ll` is specified) and that the upper limit is the maximum (if `ul` is specified).

`mle` requests that the conditional maximum-likelihood estimator be used. This is the default.

`constraints(`*constraints*`)`; see [R] **estimation options**.

SE/Robust

`vce(`*vcetype*`)` specifies the type of standard error reported, which includes types that are derived from asymptotic theory (`oim`, `opg`), that are robust to some kinds of misspecification (`robust`), that allow for intragroup correlation (`cluster` *clustvar*), and that use bootstrap or jackknife methods (`bootstrap`, `jackknife`); see [R] ***vce_option***.

Reporting

`level(`*#*`)`; see [R] **estimation options**.

`first` requests that the parameters for the reduced-form equations showing the relationships between the endogenous variables and instruments be displayed. For the two-step estimator, `first` shows the first-stage regressions. For the maximum likelihood estimator, these parameters are estimated jointly with the parameters of the tobit equation. The default is not to show these parameter estimates.

`nocnsreport`; see [R] **estimation options**.

display_options: `noomitted`, `vsquish`, `noemptycells`, `baselevels`, `allbaselevels`, `nofvlabel`, `fvwrap(`*#*`)`, `fvwrapon(`*style*`)`, `cformat(`*%fmt*`)`, `pformat(`*%fmt*`)`, `sformat(`*%fmt*`)`, and `nolstretch`; see [R] **estimation options**.

Maximization

maximize_options: `difficult`, `technique(`*algorithm_spec*`)`, `iterate(`*#*`)`, [`no`]`log`, `trace`, `gradient`, `showstep`, `hessian`, `showtolerance`, `tolerance(`*#*`)`, `ltolerance(`*#*`)`, `nrtolerance(`*#*`)`, `nonrtolerance`, and `from(`*init_specs*`)`; see [R] **maximize**. This model's likelihood function can be difficult to maximize, especially with multiple endogenous variables. The `difficult` and `technique(bfgs)` options may be helpful in achieving convergence.

Setting the optimization type to `technique(bhhh)` resets the default *vcetype* to `vce(opg)`.

The following option is available with `ivtobit` but is not shown in the dialog box:

`coeflegend`; see [R] **estimation options**.

Options for two-step estimator

Model

`twostep` is required and requests that Newey's (1987) efficient two-step estimator be used to obtain the coefficient estimates.

`ll(#)` and `ul(#)` indicate the lower and upper limits for censoring, respectively. You may specify one or both. Observations with *depvar* $\leq$ `ll()` are left-censored; observations with *depvar* $\geq$ `ul()` are right-censored; and remaining observations are not censored. You do not have to specify the censoring values at all. It is enough to type `ll`, `ul`, or both. When you do not specify a censoring value, `ivtobit` assumes that the lower limit is the minimum observed in the data (if `ll` is specified) and that the upper limit is the maximum (if `ul` is specified).

SE

`vce(`*vcetype*`)` specifies the type of standard error reported, which includes types that are derived from asymptotic theory (`twostep`) and that use bootstrap or jackknife methods (`bootstrap`, `jackknife`); see [R] ***vce_option***.

Reporting

`level(#)`; see [R] **estimation options**.

`first` requests that the parameters for the reduced-form equations showing the relationships between the endogenous variables and instruments be displayed. For the two-step estimator, `first` shows the first-stage regressions. For the maximum likelihood estimator, these parameters are estimated jointly with the parameters of the tobit equation. The default is not to show these parameter estimates.

display_options: `noomitted`, `vsquish`, `noemptycells`, `baselevels`, `allbaselevels`, `nofvlabel`, `fvwrap(#)`, `fvwrapon(`*style*`)`, `cformat(%`*fmt*`)`, `pformat(%`*fmt*`)`, `sformat(%`*fmt*`)`, and `nolstretch`; see [R] **estimation options**.

The following option is available with `ivtobit` but is not shown in the dialog box:

`coeflegend`; see [R] **estimation options**.

Remarks and examples

stata.com

`ivtobit` fits models with censored dependent variables and endogenous regressors. You can use it to fit a tobit model when you suspect that one or more of the regressors is correlated with the error term. `ivtobit` is to tobit what `ivregress` is to linear regression analysis; see [R] **ivregress** for more information.

Formally, the model is

$$\begin{aligned} y_{1i}^* &= \boldsymbol{y}_{2i}\boldsymbol{\beta} + \boldsymbol{x}_{1i}\boldsymbol{\gamma} + u_i \\ \boldsymbol{y}_{2i} &= \boldsymbol{x}_{1i}\boldsymbol{\Pi}_1 + \boldsymbol{x}_{2i}\boldsymbol{\Pi}_2 + \boldsymbol{v}_i \end{aligned}$$

where $i = 1, \ldots, N$; $\boldsymbol{y}_{2i}$ is a $1 \times p$ vector of endogenous variables; $\boldsymbol{x}_{1i}$ is a $1 \times k_1$ vector of exogenous variables; $\boldsymbol{x}_{2i}$ is a $1 \times k_2$ vector of additional instruments; and the equation for $\boldsymbol{y}_{2i}$ is written in reduced form. By assumption $(u_i, \boldsymbol{v}_i) \sim \mathrm{N}(\mathbf{0})$. $\boldsymbol{\beta}$ and $\boldsymbol{\gamma}$ are vectors of structural parameters, and $\boldsymbol{\Pi}_1$ and $\boldsymbol{\Pi}_2$ are matrices of reduced-form parameters. We do not observe y_{1i}^*; instead, we observe

$$y_{1i} = \begin{cases} a & y_{1i}^* < a \\ y_{1i}^* & a \leq y_{1i}^* \leq b \\ b & y_{1i}^* > b \end{cases}$$

The order condition for identification of the structural parameters is that $k_2 \geq p$. Presumably, $\boldsymbol{\Sigma}$ is not block diagonal between u_i and $\boldsymbol{v}_i$; otherwise, $\boldsymbol{y}_{2i}$ would not be endogenous.

❑ Technical note

This model is derived under the assumption that $(u_i, \boldsymbol{v}_i)$ is independent and identically distributed multivariate normal for all i. The `vce(cluster` *clustvar*`)` option can be used to control for a lack of independence. As with the standard tobit model without endogeneity, if u_i is heteroskedastic, point estimates will be inconsistent.

❑

▷ Example 1

Using the same dataset as in [R] **ivprobit**, we now want to estimate women's incomes. In our hypothetical world, all women who choose not to work receive $10,000 in welfare and child-support payments. Therefore, we never observe incomes under $10,000: a woman offered a job with an annual wage less than that would not accept and instead would collect the welfare payment. We model income as a function of the number of years of schooling completed, the number of children at home, and other household income. We again believe that `other_inc` is endogenous, so we use `male_educ` as an instrument.

```
. use http://www.stata-press.com/data/r13/laborsup
. ivtobit fem_inc fem_educ kids (other_inc = male_educ), ll
Fitting exogenous tobit model
Fitting full model
Iteration 0:   log likelihood = -3228.4224
Iteration 1:   log likelihood = -3226.2882
Iteration 2:   log likelihood =  -3226.085
Iteration 3:   log likelihood = -3226.0845
Iteration 4:   log likelihood = -3226.0845
Tobit model with endogenous regressors            Number of obs   =        500
                                                  Wald chi2(3)    =     117.42
Log likelihood = -3226.0845                       Prob > chi2     =     0.0000

------------------------------------------------------------------------------
             |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
   other_inc |  -.9045399   .1329762    -6.80   0.000    -1.165168   -.6439114
    fem_educ |   3.272391   .3968708     8.25   0.000     2.494538    4.050243
        kids |  -3.312357   .7218628    -4.59   0.000    -4.727182   -1.897532
       _cons |   19.24735   7.372391     2.61   0.009     4.797725    33.69697
-------------+----------------------------------------------------------------
      /alpha |   .2907654   .1379965     2.11   0.035     .0202972    .5612336
        /lns |   2.874031   .0506672    56.72   0.000     2.774725    2.973337
        /lnv |   2.813383   .0316228    88.97   0.000     2.751404    2.875363
-------------+----------------------------------------------------------------
           s |   17.70826    .897228                      16.03422    19.55707
           v |   16.66621   .5270318                      15.66461    17.73186
------------------------------------------------------------------------------
Instrumented:  other_inc
Instruments:   fem_educ kids male_educ
------------------------------------------------------------------------------
Wald test of exogeneity (/alpha = 0): chi2(1) =     4.44  Prob > chi2 = 0.0351
  Obs. summary:        272  left-censored observations at fem_inc<=10
                       228     uncensored observations
                         0 right-censored observations
```

Because we did not specify `mle` or `twostep`, `ivtobit` used the maximum likelihood estimator by default. `ivtobit` fits a tobit model, ignoring endogeneity, to get starting values for the full model. The header of the output contains the maximized log likelihood, the number of observations, and a

Wald statistic and p-value for the test of the hypothesis that all the slope coefficients are jointly zero. At the end of the output, we see a count of the censored and uncensored observations.

Near the bottom of the output is a Wald test of the exogeneity of the instrumented variables. If the test statistic is not significant, there is not sufficient information in the sample to reject the null hypothesis of no endogeneity. Then the point estimates from `ivtobit` are consistent, although those from `tobit` are likely to have smaller standard errors.

◁

Various two-step estimators have also been proposed for the endogenous tobit model, and Newey's (1987) minimum chi-squared estimator is available with the `twostep` option.

▷ Example 2

Refitting our labor-supply model with the two-step estimator yields

```
. ivtobit fem_inc fem_educ kids (other_inc = male_educ), ll twostep
Two-step tobit with endogenous regressors         Number of obs    =       500
                                                  Wald chi2(3)     =    117.38
                                                  Prob > chi2      =    0.0000

------------------------------------------------------------------------------
             |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
   other_inc |  -.9045397   .1330015    -6.80   0.000    -1.165218   -.6438616
    fem_educ |    3.27239   .3969399     8.24   0.000     2.494402    4.050378
        kids |  -3.312356   .7220066    -4.59   0.000    -4.727463   -1.897249
       _cons |   19.24735    7.37392     2.61   0.009     4.794728    33.69997
------------------------------------------------------------------------------
Instrumented:  other_inc
Instruments:   fem_educ kids male_educ
------------------------------------------------------------------------------
Wald test of exogeneity:      chi2(1) =     4.64          Prob > chi2 = 0.0312
  Obs. summary:         272  left-censored observations at fem_inc<=10
                        228     uncensored observations
                          0 right-censored observations
```

All the coefficients have the same signs as their counterparts in the maximum likelihood model. The Wald test at the bottom of the output confirms our earlier finding of endogeneity.

◁

❑ Technical note

In the tobit model with endogenous regressors, we assume that $(u_i, \boldsymbol{v}_i)$ is multivariate normal with covariance matrix

$$\mathrm{Var}(u_i, \boldsymbol{v}_i) = \boldsymbol{\Sigma} = \begin{bmatrix} \sigma_u^2 & \boldsymbol{\Sigma}_{21}' \\ \boldsymbol{\Sigma}_{21} & \boldsymbol{\Sigma}_{22} \end{bmatrix}$$

Using the properties of the multivariate normal distribution, $\mathrm{Var}(u_i|\boldsymbol{v}_i) \equiv \sigma_{u|v}^2 = \sigma_u^2 - \boldsymbol{\Sigma}_{21}'\boldsymbol{\Sigma}_{22}^{-1}\boldsymbol{\Sigma}_{21}$. Calculating the marginal effects on the conditional expected values of the observed and latent dependent variables and on the probability of censoring requires an estimate of σ_u^2. The two-step estimator identifies only $\sigma_{u|v}^2$, not σ_u^2, so only the linear prediction and its standard error are available after you have used the `twostep` option. However, unlike the two-step probit estimator described in [R] **ivprobit**, the two-step tobit estimator does identify $\boldsymbol{\beta}$ and $\boldsymbol{\gamma}$. See Wooldridge (2010, 683–684) for more information.

❑

Stored results

ivtobit, mle stores the following in e():

Scalars

e(N)	number of observations
e(N_unc)	number of uncensored observations
e(N_lc)	number of left-censored observations
e(N_rc)	number of right-censored observations
e(llopt)	contents of ll()
e(ulopt)	contents of ul()
e(k)	number of parameters
e(k_eq)	number of equations in e(b)
e(k_eq_model)	number of equations in overall model test
e(k_aux)	number of auxiliary parameters
e(k_dv)	number of dependent variables
e(df_m)	model degrees of freedom
e(ll)	log likelihood
e(N_clust)	number of clusters
e(endog_ct)	number of endogenous regressors
e(p)	model Wald p-value
e(p_exog)	exogeneity test Wald p-value
e(chi2)	model Wald χ^2
e(chi2_exog)	Wald χ^2 test of exogeneity
e(rank)	rank of e(V)
e(ic)	number of iterations
e(rc)	return code
e(converged)	1 if converged, 0 otherwise

Macros

e(cmd)	ivtobit
e(cmdline)	command as typed
e(depvar)	name of dependent variable
e(instd)	instrumented variables
e(insts)	instruments
e(wtype)	weight type
e(wexp)	weight expression
e(title)	title in estimation output
e(clustvar)	name of cluster variable
e(chi2type)	Wald; type of model χ^2 test
e(vce)	*vcetype* specified in vce()
e(vcetype)	title used to label Std. Err.
e(method)	ml
e(opt)	type of optimization
e(which)	max or min; whether optimizer is to perform maximization or minimization
e(ml_method)	type of ml method
e(user)	name of likelihood-evaluator program
e(technique)	maximization technique
e(properties)	b V
e(predict)	program used to implement predict
e(footnote)	program used to implement the footnote display
e(marginsok)	predictions allowed by margins
e(asbalanced)	factor variables fvset as asbalanced
e(asobserved)	factor variables fvset as asobserved

Matrices	
e(b)	coefficient vector
e(Cns)	constraints matrix
e(ilog)	iteration log (up to 20 iterations)
e(gradient)	gradient vector
e(Sigma)	$\widehat{\Sigma}$
e(V)	variance–covariance matrix of the estimators
e(V_modelbased)	model-based variance
Functions	
e(sample)	marks estimation sample

`ivtobit, twostep` stores the following in `e()`:

Scalars	
e(N)	number of observations
e(N_unc)	number of uncensored observations
e(N_lc)	number of left-censored observations
e(N_rc)	number of right-censored observations
e(llopt)	contents of `ll()`
e(ulopt)	contents of `ul()`
e(df_m)	model degrees of freedom
e(df_exog)	degrees of freedom for χ^2 test of exogeneity
e(p)	model Wald p-value
e(p_exog)	exogeneity test Wald p-value
e(chi2)	model Wald χ^2
e(chi2_exog)	Wald χ^2 test of exogeneity
e(rank)	rank of `e(V)`
Macros	
e(cmd)	`ivtobit`
e(cmdline)	command as typed
e(depvar)	name of dependent variable
e(instd)	instrumented variables
e(insts)	instruments
e(wtype)	weight type
e(wexp)	weight expression
e(chi2type)	`Wald`; type of model χ^2 test
e(vce)	*vcetype* specified in `vce()`
e(vcetype)	title used to label Std. Err.
e(method)	`twostep`
e(properties)	`b V`
e(predict)	program used to implement `predict`
e(footnote)	program used to implement the footnote display
e(marginsok)	predictions allowed by `margins`
e(asbalanced)	factor variables `fvset` as `asbalanced`
e(asobserved)	factor variables `fvset` as `asobserved`
Matrices	
e(b)	coefficient vector
e(Cns)	constraints matrix
e(V)	variance–covariance matrix of the estimators
e(V_modelbased)	model-based variance
Functions	
e(sample)	marks estimation sample

Methods and formulas

The estimation procedure used by `ivtobit` is similar to that used by `ivprobit`. For compactness, we write the model as

$$y_{1i}^* = \boldsymbol{z}_i\boldsymbol{\delta} + u_i \tag{1a}$$
$$\boldsymbol{y}_{2i} = \boldsymbol{x}_i\boldsymbol{\Pi} + \boldsymbol{v}_i \tag{1b}$$

where $\boldsymbol{z}_i = (\boldsymbol{y}_{2i}, \boldsymbol{x}_{1i})$, $\boldsymbol{x}_i = (\boldsymbol{x}_{1i}, \boldsymbol{x}_{2i})$, $\boldsymbol{\delta} = (\boldsymbol{\beta}', \boldsymbol{\gamma}')'$, and $\boldsymbol{\Pi} = (\boldsymbol{\Pi}_1', \boldsymbol{\Pi}_2')'$. We do not observe y_{1i}^*; instead, we observe

$$y_{1i} = \begin{cases} a & y_{1i}^* < a \\ y_{1i}^* & a \le y_{1i}^* \le b \\ b & y_{1i}^* > b \end{cases}$$

$(u_i, \boldsymbol{v}_i)$ is distributed multivariate normal with mean zero and covariance matrix

$$\boldsymbol{\Sigma} = \begin{bmatrix} \sigma_u^2 & \boldsymbol{\Sigma}_{21}' \\ \boldsymbol{\Sigma}_{21} & \boldsymbol{\Sigma}_{22} \end{bmatrix}$$

Using the properties of the multivariate normal distribution, we can write $u_i = \boldsymbol{v}_i'\boldsymbol{\alpha} + \epsilon_i$, where $\boldsymbol{\alpha} = \boldsymbol{\Sigma}_{22}^{-1}\boldsymbol{\Sigma}_{21}$; $\epsilon_i \sim \mathrm{N}(0; \sigma_{u|v}^2)$, where $\sigma_{u|v}^2 = \sigma_u^2 - \boldsymbol{\Sigma}_{21}'\boldsymbol{\Sigma}_{22}^{-1}\boldsymbol{\Sigma}_{21}$; and ϵ_i is independent of $\boldsymbol{v}_i$, $\boldsymbol{z}_i$, and $\boldsymbol{x}_i$.

The likelihood function is straightforward to derive because we can write the joint density $f\left(y_{1i}, \boldsymbol{y}_{2i}|\boldsymbol{x}_i\right)$ as $f\left(y_{1i}|\boldsymbol{y}_{2i}, \boldsymbol{x}_i\right) f\left(\boldsymbol{y}_{2i}|\boldsymbol{x}_i\right)$. With one endogenous regressor,

$$\ln f(y_{2i}|\boldsymbol{x}_i) = -\frac{1}{2}\left\{\ln 2\pi + \ln\sigma_v^2 + \frac{(y_{2i} - \boldsymbol{x}_i\boldsymbol{\Pi})^2}{\sigma_v^2}\right\}$$

and

$$\ln f(y_{1i}|y_{2i}, \boldsymbol{x}_i) = \begin{cases} \ln\left\{1 - \Phi\left(\frac{m_i - a}{\sigma_{u|v}}\right)\right\} & y_{1i} = a \\ -\frac{1}{2}\left\{\ln 2\pi + \ln\sigma_{u|v}^2 + \frac{(y_{1i} - m_i)^2}{\sigma_{u|v}^2}\right\} & a < y_{1i} < b \\ \ln\Phi\left(\frac{m_i - b}{\sigma_{u|v}}\right) & y_{1i} = b \end{cases}$$

where

$$m_i = \boldsymbol{z}_i\boldsymbol{\delta} + \alpha\left(y_{2i} - \boldsymbol{x}_i\boldsymbol{\Pi}\right)$$

and $\Phi(\cdot)$ is the normal distribution function so that the log likelihood for observation i is

$$\ln L_i = w_i\left\{\ln f(y_{1i}|y_{2i}, \boldsymbol{x}_i) + \ln f(y_{2i}|\boldsymbol{x}_i)\right\}$$

where w_i is the weight for observation i or one if no weights were specified. Instead of estimating $\sigma_{u|v}$ and σ_v directly, we estimate $\ln\sigma_{u|v}$ and $\ln\sigma_v$.

For multiple endogenous regressors, we have

$$\ln f(\boldsymbol{y}_{2i}|\boldsymbol{x}_i) = -\frac{1}{2}\left(\ln 2\pi + \ln|\boldsymbol{\Sigma}_{22}| + \boldsymbol{v}_i'\boldsymbol{\Sigma}_{22}^{-1}\boldsymbol{v}_i\right)$$

and $\ln f(y_{1i}|\boldsymbol{y}_{2i}, \boldsymbol{x}_i)$ is the same as before, except that now

$$m_i = \boldsymbol{z}_i\boldsymbol{\delta} + (\boldsymbol{y}_{2i} - \boldsymbol{x}_i\boldsymbol{\Pi})\boldsymbol{\Sigma}_{22}^{-1}\boldsymbol{\Sigma}_{21}$$

Instead of maximizing the log-likelihood function with respect to $\boldsymbol{\Sigma}$, we maximize with respect to the Cholesky decomposition $\boldsymbol{S}$ of $\boldsymbol{\Sigma}$; that is, there exists a lower triangular matrix $\boldsymbol{S}$ such that $\boldsymbol{S}\boldsymbol{S}' = \boldsymbol{\Sigma}$. This maximization ensures that $\boldsymbol{\Sigma}$ is positive definite, as a covariance matrix must be. Let

$$\boldsymbol{S} = \begin{bmatrix} s_{11} & 0 & 0 & \dots & 0 \\ s_{21} & s_{22} & 0 & \dots & 0 \\ s_{31} & s_{32} & s_{33} & \dots & 0 \\ \vdots & \vdots & \vdots & \ddots & \vdots \\ s_{p+1,1} & s_{p+1,2} & s_{p+1,3} & \dots & s_{p+1,p+1} \end{bmatrix}$$

With maximum likelihood estimation, this command supports the Huber/White/sandwich estimator of the variance and its clustered version using `vce(robust)` and `vce(cluster` *clustvar*`)`, respectively. See [P] **_robust**, particularly *Maximum likelihood estimators* and *Methods and formulas*.

The maximum likelihood version of `ivtobit` also supports estimation with survey data. For details on VCEs with survey data, see [SVY] **variance estimation**.

The two-step estimates are obtained using Newey's (1987) minimum chi-squared estimator. The procedure is identical to the one described in [R] **ivprobit**, except that `tobit` is used instead of `probit`.

Acknowledgments

The two-step estimator is based on the `tobitiv` command written by Jonah Gelbach of the Department of Economics at Yale University and the `ivtobit` command written by Joe Harkness of the Institute of Policy Studies at Johns Hopkins University.

References

Finlay, K., and L. M. Magnusson. 2009. Implementing weak-instrument robust tests for a general class of instrumental-variables models. *Stata Journal* 9: 398–421.

Miranda, A., and S. Rabe-Hesketh. 2006. Maximum likelihood estimation of endogenous switching and sample selection models for binary, ordinal, and count variables. *Stata Journal* 6: 285–308.

Newey, W. K. 1987. Efficient estimation of limited dependent variable models with endogenous explanatory variables. *Journal of Econometrics* 36: 231–250.

Wooldridge, J. M. 2010. *Econometric Analysis of Cross Section and Panel Data*. 2nd ed. Cambridge, MA: MIT Press.

Also see

[R] **ivtobit postestimation** — Postestimation tools for ivtobit

[R] **gmm** — Generalized method of moments estimation

[R] **ivprobit** — Probit model with continuous endogenous regressors

[R] **ivregress** — Single-equation instrumental-variables regression

[R] **regress** — Linear regression

[R] **tobit** — Tobit regression

[SVY] **svy estimation** — Estimation commands for survey data

[XT] **xtintreg** — Random-effects interval-data regression models

[XT] **xttobit** — Random-effects tobit models

[U] **20 Estimation and postestimation commands**